Solutions to Practice Problems 4.3

1. **(a)** Valid. These items are redundant because 1 and 4 are just special cases of `Is < 10`. However, this makes no difference in Visual Basic.

   (b) Valid. These items are contradictory. However, Visual Basic looks at them one at a time until it finds an item containing the value of the selector. The action following this Case clause will always be carried out.

   (c) Not valid. It should be `Case 2`.

2. Yes. However, the program on the right is clearer and therefore preferable.

4.4 Input via User Selections

Programs frequently ask the user to make selections from lists of options. In the questionnaire in Fig. 4.8, students can select their major from a list box, their year from a set of radio buttons, and their computer languages studied from a set of check boxes. After the selections have been made, the user clicks on the *Record Data* button to process the information. The set of radio buttons and set of check boxes are each contained in a group box control. The titles sunk into the tops of the group boxes are *Year* and *Languages Studied*. After selections are made from the three sets of choices, decision structures can be used to process the information. Let's consider the four types of controls in Fig. 4.8 one at a time.



FIGURE 4.8 Selection controls.

Using a List Box for Input

The easiest way to populate a list box with items is to place the items into the list box's String Collection Editor at design time. When you click on the list box's *Tasks* button and then click on *Edit Items* (Fig. 4.9), the String Collection Editor appears. Figure 4.10 shows the String Collection Editor filled with the items from Fig. 4.8. Three ways to fill the String Collection Editor are as follows:

1. Type the items directly into the String Collection Editor.
2. Copy a list of items from any text editor (such as Notepad or Word) with Ctrl + C and paste the list into the String Collection Editor with Ctrl + V.
3. Copy a column of data from a spreadsheet program (such as Excel) and paste it into the String Collection Editor.

When the user clicks on an item at run time, that item is highlighted, and the value of lstBox.Text is that item represented as a string.



FIGURE 4.9 Click on *Edit Items* to invoke the String Collection Editor.



FIGURE 4.10 String Collection Editor.

Example 1 The following program uses a list of months. The months can be typed directly into the String Collection Editor. An alternate way to obtain the list of months is to generate it with Excel. To do so, type January into a cell of an Excel spreadsheet [see Fig. 4.11(a)], click on the cell, drag its fill handle down to create the other months [see Fig. 4.11(b)], and press Ctrl + C. Then the list of twelve months can be pasted into the String Collection Editor with Ctrl + V.

```
Private Sub btnDetermine_Click(...) Handles btnDetermine.Click
  Dim daysInMonth As String
  Select Case lstMonths.Text
    Case "September", "April", "June", "November"
      daysInMonth = "30"
    Case "February"
      daysInMonth = "28 or 29"
```

```
    Case Else
      daysInMonth = "31"
  End Select
  txtDays.Text = daysInMonth
End Sub
```

[Run, click on a month, and then click on the button.]





(a) (b)

FIGURE 4.11 **Creating a list of months with Excel.**

Example 2 The following program presents a list of famous movie lines spoken by leading male actors. After the user makes a selection and clicks on the button, the actor is identified. The numbering of the lines allows the Select Case block to be simplified.

```
Private Sub btnDetermine_Click(...) Handles btnDetermine.Click
  Dim actor As String = ""
  Select Case lstLines.Text.Substring(0, 1)
    Case "1", "2"
      actor = "Marlon Brando"
    Case "3", "4"
```

```
            actor = "Humphrey Bogart"
        Case "5", "6"
            actor = "Harrison Ford"
        Case "7"
            actor = "Leonardo DiCaprio"
        Case "8"
            actor = "Roy Scheider"
    End Select
    txtActor.Text = actor
End Sub
```

[Run, click on one of the lines in the list box, and then click on the button.]



When no item in a list box is highlighted, the value of lstBox.SelectedItem is *Nothing*. Also, the statement `lstBox.SelectedItem = Nothing` deselects any item that has been selected. By default, only one item at a time can be selected in a list box.

Group Box Control

Group boxes are passive objects that contain related sets of controls. In Fig. 4.7, the group box titled *Year* contains four radio button controls and the group box titled *Languages Studied* contains six check boxes. Clusters of radio buttons are almost always contained in group boxes. Actually, any set of controls can be placed into a group box for visual effect.

You rarely write event procedures for group boxes. When you move a group box, the controls inside it follow as a unit. Therefore, the controls are said to be **attached** to the group box. If you hide a group box, the attached controls will be hidden as well. To attach a control to a group box, just create the control any way you like and drag it inside the group box. The standard prefix for the name of a group box is *grp*. The title sunk into the top of a group box's border is set with the Text property.

Using Radio Buttons for Input

Radio buttons allow the user to make a single choice from among several options. The name "radio button" comes from a reference to the first car radios, which had buttons that pressed in. Pressing one button would move the dial to a preset station and would raise any other button that was depressed.

Normally, a collection of several radio buttons is attached to a group box. Each button consists of a small circle accompanied by a caption that is set with the Text property. (As with ordinary buttons, an ampersand can be used to create an access key for a radio button.) When a circle or its accompanying caption is clicked, a solid dot appears in the circle and the button is said to be on. At most one radio button in a group can be on at any one time. Therefore, if one button is on and another button in the group is clicked, the first button will be turned off. The standard prefix for the name of a radio button is *rad*. A single form can have several groups of radio buttons.

The Checked property of a radio button tells if the button is on or off. The condition

```
radButton.Checked
```

is true when radButton is on and false when radButton is off. The statement

```
radButton.Checked = True
```

turns on radButton and turns off all other buttons in its group. The statement

```
radButton.Checked = False
```

turns off radButton and has no effect on the other buttons in its group.

Example 3 The following program displays the admission fee to an event. After the user clicks on a radio button and clicks on the *Determine Fee* button, the fee is displayed in a text box. The Else clause in the If block handles the case where no selection was made.



OBJECT	PROPERTY	SETTING
frmFee	Text	Admission Fee
grpAge	Text	Age
radChild	Text	child (< 6)
radMinor	Text	minor (6-17)
radAdult	Text	adult (18-64)
radSenior	Text	senior (65+)
btnDetermine	Text	Determine Fee
lblFee	Text	Fee:
txtFee	ReadOnly	True

```
Private Sub btnDetermine_Click(...) Handles btnDetermine.Click
  If radChild.Checked Then
    txtFee.Text = FormatCurrency(0)
  ElseIf radMinor.Checked Then
    txtFee.Text = FormatCurrency(5)
  ElseIf radAdult.Checked Then
    txtFee.Text = FormatCurrency(10)
  ElseIf radSenior.Checked Then
    txtFee.Text = FormatCurrency(7.5)
  Else
    MessageBox.Show("You must make a selection.")
  End If
End Sub
```

[Run, click on a radio button, and click on the *Determine Fee* button]



Using Check Boxes for Input

A **check box**, which consists of a small square and a caption (set with the Text property), presents the user with a yes/no choice. The form in Fig. 4.7 contains six check box controls. The Checked property of a check box has the value False when the square is empty and True when the square is checked. At run time, the user clicks on the square (or its accompanying caption) to toggle between the unchecked and checked states.

Example 4 The following program calculates the monthly cost of a company health plan. The user checks the desired plans, and then clicks on the button to calculate the total cost.



OBJECT	PROPERTY	SETTING
frmMenu	Text	Benefits Menu
chkDrug	Text	Prescription Drug Plan ($39.15)
chkDental	Text	Dental Plan ($10.81)
chkVision	Text	Vision Plan ($2.25)
chkMedical	Text	Medical Plan ($55.52)
btnDetermine	Text	Determine Total Monthly Cost
lblTotal	Text	Total monthly cost:
txtTotal	ReadOnly	True

```
Private Sub btnDetermine_Click(...) Handles btnDetermine.Click
  Dim sum As Double = 0
  If chkDrug.Checked Then
    sum += 39.15
  End If
  If chkDental.Checked Then
    sum += 10.81
  End If
  If chkVision.Checked Then
    sum += 2.25
  End If
  If chkMedical.Checked Then
    sum += 55.52
  End If
  txtTotal.Text = FormatCurrency(sum)
End Sub
```

[Run, select plans, and then click on the button.]



When a check box has the focus, the spacebar can be used to check (or uncheck) the box. In addition, the state of a check box can be toggled from the keyboard without first setting the focus to the check box if you create an access key for the check box by including an ampersand in the Text property. (At run time, access keys appear underlined after the Alt key is pressed.) For instance, if the Text property for the Dental Plan in Example 4 is set as "&Dental Plan", then the user can check (or uncheck) the box by pressing Alt + D.

Events Raised by Selections

In most real-life programs, the user is asked to make selections from several different controls and then click on a button in order to process the information. Such is the case with the Visual Basic Font selection window shown in Fig. 4.12, where the *OK* button is clicked after the selections have been made. Sometimes however, you would like to process information as soon as an item in a list box, a radio button, or a check box is clicked. Visual Basic provides event procedures for immediate processing.



FIGURE 4.12 **Font selection window.**

When you click on an item in a list box, the **SelectedIndexChanged event** is raised. When you click on a radio button or a check box, the **CheckedChanged event** is raised. These events are the default events for their controls.

Example 5 The following variation of Example 4 keeps a running total of the monthly cost of the benefits. The amount in the text box is increased each time a check box becomes checked and is decreased each time a check box becomes unchecked. To create the header for the event procedure we double-clicked on the first check box, changed the name of the event procedure from chkDrug_CheckedChanged to checkBox_CheckedChanged, and added three additional events following the keyword Handles.



OBJECT	PROPERTY	SETTING
frmMenu	Text	Benefits Menu
chkDrug	Text	Prescription Drug Plan ($39.15)
chkDental	Text	Dental Plan ($10.81)
chkVision	Text	Vision Plan ($2.25)
chkMedical	Text	Medical Plan ($55.52)
lblTotal	Text	Total monthly cost:
txtTotal	ReadOnly	True

```
Private Sub checkBox_CheckedChanged(...) Handles _
                        chkDrug.CheckedChanged, chkDental.CheckedChanged,
                        chkVision.CheckedChanged, chkMedical.CheckedChanged
  Dim sum As Double
  If chkDrug.Checked Then
    sum += 39.15
  End If
  If chkDental.Checked Then
    sum += 10.81
  End If
  If chkVision.Checked Then
    sum += 2.25
  End If
  If chkMedical.Checked Then
    sum += 55.52
  End If
  txtTotal.Text = FormatCurrency(sum)
End Sub
```

Comments

1. Both list boxes and radio buttons can be used to select a single item from a list of options. As a rule of thumb, radio buttons should be used with short lists (at most seven options) and list boxes should be used with long lists.
2. When the user clicks on a checked check box, it becomes unchecked. Such is not the case with a radio button. A radio button can only be unchecked with code or by clicking on another radio button.

Practice Problems 4.4

1. What is the difference between a set of check boxes attached to a group box and a set of radio buttons attached to a group box?
2. Suppose a form contains two sets of radio buttons. Why is it essential for the sets to be contained in separate group boxes? Do the same concerns apply to sets of check boxes?
3. Suppose a group box contains a set of radio buttons. Give two ways to guarantee that the user will select one of the radio buttons.

EXERCISES 4.4

In Exercises 1 through 8, determine the effect of setting the property to the value shown.

1. `GroupBox1.Text = "Income"`
2. `CheckBox1.Checked = True`
3. `CheckBox1.Checked = False`
4. `CheckBox1.Text = "&Vanilla"`
5. `RadioButton1.Checked = False`
6. `txtOutput.Text = lstBox.Text`
7. `RadioButton1.Text = "Clear &All"`
8. `RadioButton1.Checked = True`

In Exercises 9 through 12, write one or more lines of code to carry out the task.

9. Set the caption for RadioButton1 to "Yes".
10. Clear the small rectangular box of CheckBox1.
11. Guarantee that CheckBox1 is checked.
12. Turn off RadioButton2.

In Exercises 13 and 14, determine the state of the two radio buttons after Button1 is clicked.

13.
```
Private Sub Button1_Click(...) Handles Button1.Click
    RadioButton1.Checked = True
    RadioButton2.Checked = True
End Sub
```

14.
```
Private Sub Button1_Click(...) Handles Button1.Click
    RadioButton1.Checked = False
    RadioButton2.Checked = False
End Sub
```

15. Suppose that a group box has two radio buttons attached to it. If the statement

```
GroupBox1.Visible = False
```

is executed, will the radio buttons also vanish? Test your answer.

16. Create a form with two group boxes, each having two radio buttons attached to it. Run the program and confirm that the two pairs of radio buttons operate independently of each other.

17. A computer dealer offers two basic computers, the Deluxe ($1000) and the Super ($1500). The customer can order any of the following additional options: upgraded video card ($200), internal modem plus Wi-Fi ($30), or 1 GB of added memory ($120). Write a program that computes the cost of the computer system selected. See Fig. 4.13.

18. Write a program to book an airline flight. See Fig. 4.14 on the next page. If the same airport is selected from the two list boxes, the user should be informed immediately that the departure and arrival airports must be different. If no airport has been selected from one or both of the list boxes when the button is clicked, then the user should be told what information must be supplied. Use message boxes to inform the user of problems.



FIGURE 4.13 Possible outcome of Exercise 17.



FIGURE 4.14 Possible outcome of Exercise 18.

19. Write a program that allows you to vote for one of two presidential candidates. See Fig. 4.15. When the *Cast Vote* button is clicked on, the text box should display your vote. In the event that neither radio button is on, the sentence "You voted for neither." should appear in the text box.



FIGURE 4.15 Possible output for Exercise 19.

FIGURE 4.16 Possible output for Exercise 20.

20. Article II, Section 1, Clause 5 of the Constitution of the United States, states that "No Person except a natural born Citizen, or a Citizen of the United States at the time of the Adoption of this Constitution, shall be eligible to the Office of President; neither shall any Person be eligible to that Office who shall not have attained to the Age of thirty five Years, and been fourteen Years a Resident within the United States." Write a program that determines if a person is eligible to run for President of the United States in 2012. (***Note:*** A natural born citizen is a person who is born within the jurisdiction of the U.S. government. A 2012 candidate for president must achieve age 35 by Inauguration Day, 1/21/2013.) See Fig. 4.16. ***Hint:*** Use the AddYears method.

21. Write a program that uses the form in Fig. 4.8 at the beginning of the section. When the button is clicked, the program should first determine if a selection has been made from both the list box and the group of radio buttons. If not, a message box should appear telling the user which types of selections have not been made. If both selections have been made, the message "Information Processed" should be displayed.

22. Write a program that uses the form in Fig. 4.8 at the beginning of the section. When the button is clicked, the program should print the names of the computer languages studied. If no check boxes have been checked, the sentence "No languages studied." should be printed.

23. Figure 4.17 shows an item from the 2008 U.S. Individual Income Tax Return. Write a program whose form resembles item 39a. The program should look at the four check boxes and display in the large text box at the right of the item the number of boxes checked.

FIGURE 4.17 **Item 39a from Form 1040 of the 2008 U.S. Individual Income Tax Return.**

24. Write a program to specify the foreground and background colors for a label containing the words VISUAL BASIC. See Fig. 4.18. If the same color is selected from the two group boxes, the user should be informed immediately that the two colors must be different. If no color has been selected from one or both of the group boxes when the button is clicked, then the user should be told what information must be supplied. Use message boxes to inform the user of problems.



FIGURE 4.18 **Possible outcome for Exercise 24.**

FIGURE 4.19 **Possible outcome of Exercise 25.**

25. The basic monthly cost of a membership in a sport and health club is $100 for adults and $75 for seniors. Available extras cost $25 each per month. Write a program that uses the form in Fig. 4.19 to calculate a member's monthly fee. Before calculating the fee, make sure that a membership category has been selected.

Solutions to Practice Problems 4.4

1. With radio buttons, at most one button can be on at any given time, whereas several check boxes can be checked simultaneously.
2. With two sets of radio buttons, you would like to make two selections. However if the two sets are in the same group box, then at most one radio button can be on at any time.

   Since several check boxes can be checked at any time, you needn't have this concern with check boxes. However, the two sets of check boxes are usually placed in separate group boxes to improve the visual effect.
3. Method 1: Make one of the radio buttons the default radio button; that is, set its Checked property to True at design time.

   Method 2: Place the code that refers to the radio buttons in an If block that displays a message when no radio button in the group box has been checked.