4.2 If Blocks

An **If block** allows a program to decide on a course of action based on whether a certain condition is true or false.

If Block

VideoNote
If blocks

A block of the form:

```
If condition Then
  action 1
Else
  action 2
End If
```

causes the program to take *action 1* if *condition* is true and *action 2* if *condition* is false. Each action consists of one or more Visual Basic statements. After an action is taken, execution continues with the line after the If block. Figure 4.1 contains the pseudocode and flowchart for an If block.

```
If condition is true Then
  Execute action 1
Else
  Execute action 2
End If
```



FIGURE 4.1 Pseudocode and flowchart for an If block.

Example 1 The following program finds the larger of two numbers input by the user. The condition is

```
num1 > num2
```

and each action consists of a single assignment statement. With the inputs 3 and 7, the condition is false, and so the second action is taken.



OBJECT	PROPERTY	SETTING
frmMaximum	Text	Maximum
lblFirstNum	Text	First number:
txtFirstNum		
lblSecondNum	Text	Second number:
txtSecondNum		
btnFindLarger	Text	Find Larger Number
txtResult	ReadOnly	True

```
Private Sub btnFindLarger_Click(...) Handles btnFindLarger.Click
  Dim num1, num2, largerNum As Double
  num1 = CDbl(txtFirstNum.Text)
  num2 = CDbl(txtSecondNum.Text)
  If num1 > num2 Then
    largerNum = num1
  Else
    largerNum = num2
  End If
  txtResult.Text = "The larger number is " & largerNum & "."
End Sub
```

[Run, type 3 and 7 into the text boxes, and click on the button.]



Example 2 The If block in the following program has a logical operator in its condition.



OBJECT	PROPERTY	SETTING
frmQuiz	Text	A Quiz
lblQuestion	AutoSize	False
	Text	How many gallons does a ten-gallon hat hold?
txtAnswer		
btnEvaluate	Text	Evaluate Answer
txtSolution	ReadOnly	True

```
Private Sub btnEvaluate_Click(...) Handles btnEvaluate.Click
  'Evaluate answer
  Dim answer As Double
  answer = CDbl(txtAnswer.Text)
  If (answer >= 0.5) And (answer <= 1) Then
    txtSolution.Text = "Good, "
  Else
    txtSolution.Text = "No, "
  End If
  txtSolution.Text &= "it holds about 3/4 of a gallon."
End Sub
```

[Run, type 10 into the text box, and click on the button.]



The Else part of an If block can be omitted. This important type of If block appears in the next example.

Example 3 The following program offers assistance to the user before presenting a quotation.



OBJECT	PROPERTY	SETTING
frmQuotation	Text	Quotation
lblQuestion	Text	Do you know what the game of skittles is (Y/N)?
mtbAnswer	Mask	L
btnDisplay	Text	Display Quotation
txtQuote	ReadOnly	True

```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim message As String
  message = "Skittles is an old form of bowling in which a wooden " &
            "disk is used to knock down nine pins arranged in a square."
  If mtbAnswer.Text.ToUpper = "N" Then
     MessageBox.Show(message, "Definition")
  End If
  txtQuote.Text = "Life ain't all beer and skittles." &
                  " - Du Maurier (1894)"
End Sub
```

[Run, type "N" into the masked text box, and click on the button.]





[Press OK.]



[Rerun the program, type "Y" into the masked text box, click on the button, and observe that the description of the game is skipped.]

Note: Logic errors are the most difficult errors to find. A common type of logic error is the omission of the ToUpper method in Example 3.

Nested If Blocks

An action part of an If block can consist of another If block. In this situation the If blocks are said to be **nested**. Examples 4 and 5 employ nested If blocks.

Example 4 The color of the beacon light atop Boston's old John Hancock building forecasts the weather according to the following rhyme:

Steady blue, clear view.
Flashing blue, clouds due.
Steady red, rain ahead.
Flashing red, snow instead.

The following program requests a color (Blue or Red) and a mode (Steady or Flashing) as input and displays the weather forecast. Both actions associated with the main If block consist of If blocks.



OBJECT	PROPERTY	SETTING
frmWeather	Text	Weather Beacon
lblColor	Text	Color of the light (B or R):
mtbColor	Mask	L
lblMode	Text	Mode (S or F):
mtbMode	Mask	L
btnInterpret	Text	Interpret Beacon
txtForecast	ReadOnly	True

```
Private Sub btnInterpret_Click(...) Handles btnInterpret.Click
  'Interpret a weather beacon
  Dim color, mode As String
  color = mtbColor.Text
  mode = mtbMode.Text
  If mode = "S" Then
    If color = "B" Then
      txtForecast.Text = "CLEAR VIEW"
    Else    'color = "R"
      txtForecast.Text = "RAIN AHEAD"
    End If
  Else      'mode = "F"
    If color = "B" Then
      txtForecast.Text = "CLOUDS DUE"
    Else    'color = "R"
      txtForecast.Text = "SNOW AHEAD"
    End If
  End If
End Sub
```

[Run, type *R* and *S* into the masked text boxes, and click on the button.]



Example 5 The following program requests the costs and revenue for a company and displays the message "Break even" if the costs and revenue are equal; otherwise, it displays the profit or loss. The action following Else is another If block.



OBJECT	PROPERTY	SETTING
frmStatus	Text	Profit/Loss
lblCosts	Text	Costs:
txtCosts		
lblRev	Text	Revenue:
txtRev		
btnShow	Text	Show Financial Status
txtResult	ReadOnly	True

```
Private Sub btnShow_Click(...) Handles btnShow.Click
  Dim costs, revenue, profit, loss As Double
  costs = CDbl(txtCosts.Text)
```

```
    revenue = CDbl(txtRev.Text)
    If costs = revenue Then
      txtResult.Text = "Break even"
    Else
      If costs < revenue Then
        profit = revenue - costs
        txtResult.Text = "Profit is " & FormatCurrency(profit) & "."
      Else
        loss = costs - revenue
        txtResult.Text = "Loss is " & FormatCurrency(loss) & "."
      End If
    End If
  End Sub
```

[Run, type 9500 and 8000 into the text boxes, and click on the button.]



ElseIf Clauses

An extension of the If block allows for more than two possible alternatives with the inclusion of ElseIf clauses. A typical block of this type is

```
If condition 1 Then
  action 1
ElseIf condition 2 Then
  action 2
ElseIf condition 3 Then
  action 3
Else
  action 4
End If
```

Visual Basic searches for the first true condition, carries out its action, and then skips to the statement following End If. If none of the conditions are true, then Else's action is carried out. Execution then continues with the statement following the block. In general, an If block can contain any number of ElseIf clauses. As before, the Else clause is optional.

Example 6 The following program redoes Example 1 so that the program reports if the two numbers are equal.

```
Private Sub btnFindLarger_Click(...) Handles btnFindLarger.Click
  Dim num1, num2 As Double
  num1 = CDbl(txtFirstNum.Text)
  num2 = CDbl(txtSecondNum.Text)
```

```
  If (num1 > num2) Then
    txtResult.Text = "The larger number is " & num1
  ElseIf (num2 > num1) Then
    txtResult.Text = "The larger number is " & num2
  Else
    txtResult.Text = "The two numbers are equal."
  End If
End Sub
```

[Run, type 7 into both text boxes, and press the button.]



The If block in Example 7 allows us to calculate values that are not determined by a simple formula.

Example 7 The Social Security or FICA tax has two components—the Social Security benefits tax, which in 2009 is 6.2% on the first $106,800 of earnings for the year, and the Medicare tax, which is 1.45% of earnings. The following program calculates an employee's FICA tax for the current pay period.



OBJECT	PROPERTY	SETTING
frmFICA	Text	FICA Taxes
lblToDate	AutoSize	False
	Text	Total earnings for this year prior to the current pay period:
txtToDate		
lblCurrent	Text	Earnings for the current pay period:
txtCurrent		
btnCalculate	Text	Calculate FICA Taxes
lblTax	Text	FICA taxes for the current pay period:
txtTax	ReadOnly	True

```
Private Sub btnCalculate_Click(...) Handles btnCalculate.Click
  'Calculate social security benefits tax and Medicare tax
  'for a single pay period in 2009
  Const WAGE_BASE As Double = 106800   'There is no social security benefits
  '                                     tax on income above this level.
  Dim ytdEarnings, curEarnings As Double
```

```
  Dim socialSecurityBenTax, medicareTax, ficaTaxes As Double
  ytdEarnings = CDbl(txtToDate.Text)
  curEarnings = CDbl(txtCurrent.Text)
  If (ytdEarnings + curEarnings) <= WAGE_BASE Then
    socialSecurityBenTax = 0.062 * curEarnings
  ElseIf ytdEarnings < WAGE_BASE Then
    socialSecurityBenTax = 0.062 * (WAGE_BASE – ytdEarnings)
  End If
  medicareTax = 0.0145 * curEarnings
  ficaTaxes = socialSecurityBenTax + medicareTax
  txtTax.Text = FormatCurrency(ficaTaxes)
End Sub
```

[Run, type 12345.67 and 543.21 into the top two text boxes and click on the button.]



The following example illustrates the fact that when an If block contains ElseIf clauses, Visual Basic executes the action corresponding to the first condition that is satisfied and ignores all subsequent clauses—even if they also satisfy the condition.

Example 8 The following program assumes that the user will graduate (that is, has a GPA of 2 or more) and determines if the user will graduate with honors.

```
Private Sub btnDetermine_Click(...) Handles btnDetermine.Click
  Dim gpa As Double = CDbl(txtGPA.Text)
  Dim honors As String
  If gpa >= 3.9 Then
    honors = " summa cum laude."
  ElseIf gpa >= 3.6 Then
    honors = " magna cum laude."
  ElseIf gpa >= 3.3 Then
    honors = " cum laude."
  ElseIf gpa >= 2 Then
    honors = "."
  End If
  txtOutput.Text = "You graduated" & honors
End Sub
```

[Run, enter a grade point average between 2 and 4, and click on the button.]



Input Validation with If Blocks

Suppose a program calls for the user to enter a number into a text box, and then the program uses the number in a computation. If the user leaves the text box empty or enters an inappropriate number, the program will crash. The Boolean-valued function IsNumeric can be used to prevent this from happening.

Note: When *boolVal* is a Boolean value, a statement of the form

```
If boolVal = True Then
```

can be shortened to

```
If boolVal Then
```

Similarly, a statement of the form

```
If boolVal = False Then
```

can be shortened to

```
If Not boolVal Then
```

Example 9 The following program uses the function IsNumeric to guard against improper input.



OBJECT	PROPERTY	SETTING
frmAddition	Text	Addition
lblFirstNum	Text	First number:
txtFirstNum		
lblSecondNum	Text	Second number:
txtSecondNum		
btnAdd	Text	Add Numbers
lblSum	Text	Sum:
txtSum	ReadOnly	True

```
Private Sub btnAdd_Click(...) Handles btnAdd.Click
  If IsNumeric(txtFirstNum.Text) And IsNumeric(txtSecondNum.Text) Then
    txtSum.Text = CStr(CDbl(txtFirstNum.Text) + CDbl(txtSecondNum.Text))
  ElseIf Not IsNumeric(txtFirstNum.Text) Then
    If Not IsNumeric(txtSecondNum.Text) Then
      MessageBox.Show("Each text box is empty or has an improper entry.")
```

```
    Else
      MessageBox.Show("The first text box is empty or has an improper entry.")
    End If
  Else
    MessageBox.Show("The second text box is empty or has an improper entry.")
  End If
End Sub
```

[Run, leave the first text box empty, enter "two" into the second text box, and click on the button.]



If blocks also can be used to guarantee that a number input by the user is in the proper range. For instance, when the user is asked to input an exam grade, a statement such as

```
If (grade >= 0) And (grade <= 100) Then
```

can be used to guarantee that the number input is between 0 and 100.

Comments

1. Care should be taken to make If blocks easy to understand. For instance, in Fig. 4.2, the block on the left is difficult to follow and should be replaced by the clearer block on the right.

```
If cond1 Then
  If cond2 Then
    action
  End If
End If
```

```
If cond1 And cond2 Then
  action
End If
```

FIGURE 4.2 A confusing If block and an improvement.

2. In Appendix D, the section "Stepping through Programs Containing Decision Structures: Chapter 4" uses the Visual Basic debugging tools to trace the flow through an If block.

Practice Problems 4.2

1. Suppose the user is asked to input a number into txtNumber for which the square root is to be taken. Fill in the If block so that the lines of code that follow will display either the message "Number can't be negative." or the square root of the number.

```
Private Sub btnSqrt_Click(...) Handles btnSqrt.Click
  'Check reasonableness of data
  Dim num As Double
  num = CDbl(txtNumber.Text)
  If

  End If
End Sub
```

2. Improve the block

```
If a < b Then
  If c < 5 Then
    txtBox.Text = "hello"
  End If
End If
```

EXERCISES 4.2

In Exercises 1 through 12, determine the output displayed in the text box when the button is clicked.

1.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim num As Double = 4
  If num <= 9 Then
    txtOutput.Text = "Less than ten."
  Else
    If num = 4 Then
      txtOutput.Text = "Equal to four."
    End If
  End If
End Sub
```

2.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim gpa As Double = 3.49
  txtOutput.Clear()
  If gpa >= 3.5 Then
    txtOutput.Text = "Honors "
  End If
  txtOutput.Text &= "Student"
End Sub
```

3.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim a As Double = 5
  txtOutput.Clear()
  If (3 * a - 4) < 9 Then
    txtOutput.Text = "Remember, "
  End If
  txtOutput.Text &= "tomorrow is another day."
End Sub
```

4.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim change As Double = 356    'Amount of change in cents
  If change >= 100 Then
     txtOutput.Text = "Your change contains "  &
                      Int(change / 100) & " dollars."
  Else
    txtOutput.Text = "Your change contains no dollars."
  End If
End Sub
```

5.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim a as Double = 2
  Dim b As Double = 3
  Dim c As Double = 5
  If a * b < c Then
     b = 7
  Else
     b = c * a
  End If
  txtOutput.Text = CStr(b)
End Sub
```

6.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  'Cost of phone call from New York to London
  Dim length, cost As Double
  'Request the length of a phone call
  length = CDbl(InputBox("Duration of the call in minutes?"))
  'Calculate cost of phone call
  If length < 1 Then
    cost = 0.46
  Else
    cost = 0.46 + (length - 1) * 0.36
  End If
  'Display the cost of the call
  txtBox.Text = "Cost of call: " & FormatCurrency(cost)
End Sub
```

(Assume the response is *31*.)

7.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim letter As String
  letter = InputBox("Enter A, B, or C.")
  letter = letter.ToUpper
  If letter = "A" Then
    txtOutput.Text = "A, my name is Alice."
  ElseIf letter = "B" Then
    txtOutput.Text = "To be, or not to be."
  ElseIf letter = "C" Then
    txtOutput.Text = "Oh, say, can you see."
  Else
    txtOutput.Text = "Not a valid letter."
  End If
End Sub
```

(Assume the response is *B*.)

8.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim vowel As Boolean = False
  Dim ltr As String
  ltr = InputBox("Enter a letter.")
  ltr = ltr.ToUpper
  If (ltr = "A") Or (ltr = "E") Or (ltr = "I") Or
     (ltr = "O") Or (ltr = "U") Then
    vowel = True
  End If
  If vowel Then
    txtOutput.Text = ltr & " is a vowel."
  Else
    txtOutput.Text = ltr & " is not a vowel."
  End If
End Sub
```

(Assume the response is *a*.)

9.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim a As Double = 5
  If (a > 2) And ((a = 3) Or (a < 7)) Then
    txtOutput.Text = "Hi"
  End If
End Sub
```

10.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim num As Double = 5
  If num < 0 Then
    txtOutput.Text = "neg"
  Else
    If num = 0 Then
      txtOutput.Text = "zero"
    Else
      txtOutput.Text = "positive"
    End If
  End If
End Sub
```

11.
```
Private Sub btnCompute_Click(...) Handles btnCompute.Click
  Dim msg As String = "You are old enough to vote"
  Dim dateOfBirth As Date = CDate(InputBox("Enter your date of birth."))
  If dateOfBirth.AddYears(18) <= Today Then
    txtOutput.Text = msg & "."
  Else
    txtOutput.Text = msg & " in " &
        DateDiff(DateInterval.Day, Today, dateOfBirth.AddYears(18)) &
        " days."
  End If
End Sub
```

(Assume that your 18th birthday is one week away.)

12.
```
Private Sub btnCompute_Click(...) Handles btnCompute.Click
  Dim dateOfBirth As Date = CDate(InputBox("Enter your date of birth."))
```

```
  Dim nicksDateOfBirth As Date = #9/16/1992#
  If dateOfBirth < nicksDateOfBirth Then
    txtOutput.Text = "You are older than Nick."
  ElseIf dateOfBirth = nicksDateOfBirth Then
    txtOutput.Text = "You are the exact same age as Nick."
  Else
    txtOutput.Text = "You are younger than Nick."
  End If
End Sub
```

(Assume that the response is *10/25/1992*.)

In Exercises 13 through 16, identify the errors, state the type of each error (syntax, run-time, or logic), and correct the block of code.

13.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim num As Double = 0.5
  If (1 < num < 3) Then
    txtOutput.Text = "Number is between 1 and 3."
  End If
End Sub
```

14.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim num As Double = 6
  If num > 5 And < 9 Then
    txtOutput.Text = "Yes"
  Else
    txtOutput.Text = "No"
  End If
End Sub
```

15.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  Dim major As String
  major = "Computer Science"
  If major = "Business" Or "Computer Science" Then
    txtOutput.Text = "Yes"
  End If
End Sub
```

16.
```
Private Sub btnDisplay_Click(...) Handles btnDisplay.Click
  'Toggle switch from on to off and from off to on
  Dim switchOn As Boolean
  switchOn = CBool(InputBox("Enter True of False.", "The switch is on."))
  If switchOn Then
    switchOn = False
  End If
  If Not switchOn Then
    switchOn = True
  End If
  txtOutput.Text = CStr(switchOn)
End Sub
```

In Exercises 17 through 20, simplify the code.

17.
```
If (a = 2) Then
  a = 3 + a
Else
```

```
    a = 5
End If
```

18.
```
If (j = 7) Then
    b = 1
Else
    If (j <> 7) Then
        b = 2
    End If
End If
```

19.
```
message = "Is Alaska bigger than Texas and California combined?"
answer = InputBox(message)
If (answer.Substring(0, 1) = "Y") Then
    answer = "YES"
End If
If (answer.Substring(0, 1) = "y") Then
    answer = "YES"
End If
If (answer = "YES") Then
    txtOutput.Text = "Correct"
Else
    txtOutput.Text = "Wrong"
End If
```

20.
```
message = "How tall (in feet) is the Statue of Liberty?"
feet = CDbl(InputBox(message))
If (feet <= 141) Then
    lstOutput.Items.Add("Nope")
End If
If (feet > 141) Then
    If (feet < 161) Then
        lstOutput.Items.Add("Close")
    Else
        lstOutput.Items.Add("Nope")
    End If
End If
lstOutput.Items.Add("The statue is 151 feet from base to torch.")
```

21. Write a program to determine how much to tip the server in a restaurant. The tip should be 15% of the check, with a minimum of $1.

22. A bagel shop charges 75 cents per bagel for orders of less than a half-dozen bagels and 60 cents per bagel for orders of a half-dozen or more. Write a program that requests the number of bagels ordered and displays the total cost. (Test the program for orders of four bagels and a dozen bagels.)

23. A store sells widgets at 25 cents each for small orders or at 20 cents each for orders of 100 or more. Write a program that requests the number of widgets ordered and displays the total cost. (Test the program for purchases of 5 and 200 widgets.)

24. A copy center charges 5 cents per copy for the first 100 copies and 3 cents per copy for each additional copy. Write a program that requests the number of copies as input and displays the total cost. (Test the program with the quantities 25 and 125.)

25. Write a quiz program to ask "Who was the first Ronald McDonald?" The program should display "Correct." if the answer is "Willard Scott" and "Nice try." for any other answer.

26. Suppose a program has a button with the caption "Quit". Suppose also that the Name property of this button is btnQuit. Write a btnQuit_Click event procedure that gives the user a second chance before ending the program. The procedure should use an input box to request that the user confirm that the program should be terminated, and then end the program only if the user responds in the affirmative.

27. Write a program that requests three scores as input and displays the average of the two highest scores.

28. Write a program to handle a savings-account withdrawal. The program should request the current balance and the amount of the withdrawal as input and then display the new balance. If the withdrawal is greater than the original balance, the program should display "Withdrawal denied." If the new balance is less than $150, the message "Balance below $150." also should be displayed.

29. A supermarket sells apples for $1.70 per pound. Write a cashier's program that requests the number of pounds and the amount of cash tendered as input and displays the change from the transaction. If the cash is not enough, the message "I need $x.xx more." should be displayed, where $x.xx is the difference between the total cost and the cash. (Test the program with six pounds and $20, and four pounds and $10.)

30. Write a program that requests a word (with lowercase letters) as input and translates the word into pig latin. The rules for translating a word into pig latin are as follows:

(a) If the word begins with a group of consonants, move them to the end of the word and add *ay*. For instance, *chip* becomes *ipchay*.

(b) If the word begins with a vowel, add *way* to the end of the word. For instance, *else* becomes *elseway*.

31. Federal law requires that hourly employees be paid "time-and-a-half" for work in excess of 40 hours in a week. For example, if a person's hourly wage is $8 and he works 60 hours in a week, his gross pay should be

$$(40 \times 8) + (1.5 \times 8 \times (60 - 40)) = \$560$$

Write a program that requests as input the number of hours a person works in a given week and his hourly wage, and then displays his gross pay.

32. The current calendar, called the Gregorian calendar, was introduced in 1582. Every year divisible by four was declared to be a leap year, with the exception of the years ending in 00 (that is, those divisible by 100) and not divisible by 400. For instance, the years 1600 and 2000 are leap years, but 1700, 1800, and 1900 are not. Write a program that requests a year as input and states whether it is a leap year. The program should not use any variables of type Date. (Test the program on the years 2008, 2009, 1900, and 2000.)

33. Create a form with a text box and two buttons captioned Bogart and Raines. When Bogart is first pressed, the sentence "I came to Casablanca for the waters." is displayed in the text box. The next time Bogart is pressed, the sentence "I was misinformed." is displayed. When Raines is pressed, the sentence "But we're in the middle of the desert." is displayed. Run the program and then press Bogart, Raines, and Bogart to obtain a dialogue.

34. Write a program that allows the user to use a button to toggle the color of the text in a text box between black and red.

35. Write a program that allows the user ten tries to answer the question, "Which U.S. President was born on July 4?" After three incorrect guesses, the program should display the hint, "He once said, 'If you don't say anything, you won't be called upon to repeat it.'" in a message box. After seven incorrect guesses, the program should give the hint, "His nickname was 'Silent Cal.'" The number of guesses should be displayed in a text box. (See Fig. 4.3.) ***Note:*** Calvin Coolidge was born on July 4, 1872.



FIGURE 4.3 **Form for Exercise 35.**

36. Write a program that reads a test score from a text box each time a button is clicked and then displays the two highest scores whenever a second button is clicked. Use two class-level variables to track the two highest scores.

37. Write a program to play "Hide and Seek" with the name of our programming language. When the button is clicked on, the name should disappear and the caption on the button should change to "Show Name of Language." The next time the button is pressed, the name should reappear and the caption should revert to "Hide Name of Language," and so on.



OBJECT	PROPERTY	SETTING
frmHideSeek	Text	Hide and Seek
lblLanguage	Text	VB 2010
	Font.Size	26
btnDisplay	Text	Hide Name of Language

38. The flowchart in Fig. 4.5 (on the next page) calculates a person's state income tax. Write a program corresponding to the flowchart. (Test the program with taxable incomes of $15,000, $30,000, and $60,000.)

39. Rework Exercise 32 using a variable of type Date and the DateDiff function.

40. Write a program that requests your date of birth as input and tells you whether or not you are 25 years old or older. If not, the program should tell you the number of days until you will have your 25th birthday.

41. Write a program that requests your date of birth as input and tells your age. ***Hint:*** Use the DateDiff function with the DateInterval.Year option, and then use an If block to modify the result.

42. Savings accounts state an interest rate and a compounding period. If the amount deposited is P, the stated interest rate is r, and interest is compounded m times per year, then the balance in the account after one year is $P \cdot \left(1 + \frac{r}{m}\right)^m$. For instance, if $1000 is deposited at 3% interest compounded quarterly (that is, 4 times per year), then the balance after one year is

$$1000 \cdot \left(1 + \frac{.03}{4}\right)^4 = 1000 \cdot 1.0075^4 = \$1{,}030.34.$$

Interest rates with different compounding periods cannot be compared directly. The concept of APY (annual percentage yield) must be used to make the comparison. The APY for a stated interest rate r compounded m times per year is defined by

$$APY = \left(1 + \frac{r}{m}\right)^m - 1.$$

(The APY is the simple interest rate that yields the same amount of interest after one year as the compounded annual rate of interest.) Write a program to compare interest rates offered by two different banks and determine the most favorable interest rate. See Fig. 4.4.



FIGURE 4.4 Possible outcome of Exercise 42.



FIGURE 4.5 Flowchart for Exercise 38.

43. Rewrite the program in Example 8 without using ElseIf clauses. That is, the task should be carried out with a sequence of simple If blocks.

44. Rewrite the program in Example 8 so that the GPA is validated to be between 2 and 4 before the If block is executed.

Solutions to Practice Problems 4.2

1.
```
If (num < 0) Then
  MessageBox.Show("Number can't be negative.", "Input Error")
  txtNumber.Clear()
  txtNumber.Focus()
Else
  txtSquareRoot.Text = CStr(Math.Sqrt(num))
End If
```

2. The word "hello" will be displayed when (a < b) is true and (c < 5) is also true. That is, it will be displayed when both of these two conditions are true. The clearest way to write the block is

```
If (a < b) And (c < 5) Then
  txtBox.Text = "hello"
End If
```

4.3 Select Case Blocks

A Select Case block is an efficient decision-making structure that simplifies choosing among several actions. It avoids complex If constructs. If blocks make decisions based on the truth value of a condition; Select Case choices are determined by the value of an expression called a **selector**. Each possible action is preceded by a clause of the form

```
Case valueList
```

where *valueList* itemizes the values of the selector for which the action should be taken.

Example 1 The following program converts the finishing position in a horse race into a descriptive phrase. After the variable *position* is assigned a value from txtPosition, Visual Basic searches for the first Case clause whose value list contains that value and executes the succeeding statement. If the value of *position* is greater than 5, then the statement following Case Else is executed.



OBJECT	PROPERTY	SETTING
frmRace	Text	Horse Race
lblPosition	AutoSize	False
	Text	Finishing position (1, 2, 3, . . .):
txtPosition		
btnEvaluate	Text	Evaluate Position
txtOutcome	ReadOnly	True

```
Private Sub btnEvaluate_Click(...) Handles btnEvaluate.Click
  Dim position As Integer   'selector
  position = CInt(txtPosition.Text)
```